
Introduction to the Lie Algebra Package
by Yuly Billig (billig@math.carleton.ca) and Mátyás Mazzag (matyas.mazzag@gmail.

com)

Distributed under Open Source GNU AGPLv3  license; for details, see https://choosealicense.
com/licenses/agpl-3.0/

This package is available for download at:  https://github.com/yulybillig/LieAlg

Description
The LieAlg package can be used to perform calculations in infinite dimensional 
graded Lie algebras, their enveloping algebras, vertex algebras and highest weight 
modules.

Functions available are:
                    generators,  using, isgenerator , w t,  ls impli fy,  genhallmon , delta , store
                    unienv,  asimplify ,  simple,  roots ,  dotproduct ,  affine,  
                    KacMoody,  triangular ,  directsum , ideal ,  quotalgebra,   basis  
                    hwrep ,  submodule ,  quotrep ,  rep simplify ,  singularvector ,  characteristic
                    &* ,  &@,   &  ̂, & < , & < = ,  & > ,  & > = 
   
To use the LieAlg library download the files LieAlg.m (contains the interface) and
LieAlg_Hidden.m (contains supporting operations used by the library). If needed, 
these files may be generated from the sourse file by opening LieAlg_source.mws 
worksheet with MAPLE and executing it (go to Edit -> Execute -> Worksheet). In 
your own worksheet set the Maple environment variable libname so that it includes
the directory where the files are located. Activate LieAlg.m and LieAlg_Hidden.m 
using read function and then call with(LieAlg) command to load all the functions in
the library.  

To activate help on LieAlg functions, place LieAlg.help file into / l ib directory (the 
path to this directory can be found by typing libname; when you start MAPLE). For 
older versions of MAPLE, place into / l ib the file LieAlg.hdb instead (but not both!).

Initialization

restart;

#libname := "C:\\Users\\Yuly\\Documents\\MAPLE\\LIEALG", 
libname; 
libname := "/home/yuly/MAPLE/LIEALG", libname;

Add a path to directory with LieAlg files to libname. The first format is for 
Windows, second is for Linux

Load LieAlg package

Free Lie algebra

An example of a computation in a free Lie algebra.
Define the set of generators for algebra L and allow the generators() function to 
assign default weights to them. The variables used as the generators should not be 
assigned values elsewhere in the code. 

generators(F,[x,y,z]);

Simplify the expression in L.
lsimplify(y&*x);

A more comlex simplification.
a:=x&*y; b:=y&*z; lsimplify(a&*b);

Calculate the weight of the resulting expression.
wt(%);

Generate Hall-Shirshov monomials in F of the given weight. In order to construct 
this basis, we use an ordering on the set of generators.  

genhallmon([1,2,1],F);

Free Lie algebra with another grading

Define another set of generators for free Lie algebra of rank 2. We also supply the 
generators() function with our own weights for the
generators.

generators(FF,[x,y,z],[[1],[1],[1]]);

lsimplify(z&*x);

With the definition of an algebra the default algebra changes to the newly defined 
algebra. The default algebra is used

when no algebra is specified for Lie algebra computations. Without any arguments 
using() displays the algebra currently in
use along with its generators. With using(algname) we can change the default 
agebra to the desired. 

using();

Calculate the weight of the expression using the user defined weights.
wt(((x&*y)&*z)&*x);

Generating Hall Monomials in FF with weight [4].
genhallmon([4],FF);

sl(2)

An example showing how to set up the sl(2) algebra in the LieAlg pacakge.
generators(sl2,[e,h,f],[[1],[0],[-1]]);

Use the store() command to store the simplification rules for the algebra.
store([h&*e=2*e,h&*f=-2*f,e&*f=h]);

Simplify an sl(2) expression.
lsimplify((e&*f)&*(e+f));

2-Dimensional solvable algebra

generators(sol,[d,n]); store([d&*n=n]); 

lsimplify((((d&*n)&*d)&*d)&*d,sol);

lsimplify((((d&*n)&*d)&*d)&*n);
0

Direct Sums

Form a direct sum of sl(2) with a 2-dimensional solvable algebra
directsum(newalg,[sl2,sol]);

lsimplify((e+d)&*(f+n));

using();

wt(e); wt(h); wt(f);

wt(d); wt(n);

Virasoro algebra

When setting up a Lie algebra, we may use basis vectors parametrized by index 
variables. Below we use this method to define the Virasoro algebra, using 
parameters to give the Lie brackets between basis elements, as well as their 
weights. Please note that the variables used as indices should not be assigned any 
values elsewhere in the code. 

generators(Vir,[e[alpha],z],[[alpha],[0]]); store([e[alpha]&*e
[beta]=(beta-alpha)*e[alpha+beta]+delta(alpha,-beta)*((alpha^3-
alpha)/12)*z, e[alpha]&*z=0]);

Simplifying an expression in Vir.
lsimplify(e[3]&*e[-3]);

Simple finite-dimensional algebras

Defining simple algebra F4 of type F[4].
simple(F4, F[4]);

The matrix generated for F. The general naming convention is that the matrix is 
called A_algname.

eval(A_F4);

Simplification of expressions in F.
lsimplify(e[1,0,0,0]&*e[0,1,0,0]);
lsimplify(e[1,1,1,1]&*e[1,1,0,0]);
lsimplify(e[2,2,1,1]&*e[1,0,0,0]);
lsimplify(e[-1,-2,-1,-1]&*e[1,2,1,1]);

e

0

A list of positive roots (as linear combinations of simple roots) may be obtained 
using function roots.
roots(F[4]);

An invariant bilinear form may be evaluated using function dotproduct
dotproduct(e[2,4,3,2], e[-2,-4,-3,-2]);
dotproduct(e[1,0,0,0], e[0,-1,0,0]);
dotproduct(h[2],h[3]);

1
0

Quotient algebras of a free Lie algebra

To build quotient algebras first define an algebra (P in our case) with its generators.

generators(P,[x,y]);

Then we set up an ideal (Q in this example) by specifying its generators. Generators
of an ideal must be homogeneous with respect to the grading being used.

ideal(Q,[x&*(x&*(x&*y)),((x&*y)&*y)&*y]);

Define the quotient algebra S of a free Lie algebra P by its ideal Q.

quotalg(S,Q);

Simplifying an expression in S and generating basis for the quotient algebra.
lsimplify(x&*(y&*(x&*(y&*x))),S);
basis([3,2],S);
basis([4,2],S);

Kac-Moody algebras

Defining Kac-Moody algebra G. KacMoody() sets up three algebras with names
algname_plus, algname_zero and
algname_minus that have positive, zero and negative weights respectively.

C:=matrix([[2,-3],[-3,2]]);
KacMoody(G,C);

lsimplify(e[1]&*f[1]);
h

lsimplify((e[2]&*e[1])&*e[1], G);

lsimplify((((f[2]&*f[1])&*f[1])&*f[1])&*f[1], G);
0

Simplifying expressions in G and generating basis for weight [4,4] in the positive 
algebra G_plus.

lsimplify(((((e[1]&*e[2])&*e[1])&*e[2])&*f[1]),G);
lsimplify(e[1]&*(e[1]&*(e[1]&*(e[1]&*e[2]))),G);
basis([4,4],G_plus);

0

Lie algebras with triangular decomposition

A tringular decomposition of a Lie algebra is a decomposition in 3 subalgebras: 
positive subalgebra, negative subalgebra and Cartan subalgebra. Examples of such 
algebras include Kac-Moody algebras, Virasoro Lie algebra and other Kac-Moody 
like algebras. The function below is a tool for building algebras with triangular 
decompositions. To do this, one should first define the three subalgebras, and then
glue them together, specifying Lie brackets between the generators from distinct 
subalgebras. The bracket between a generator of Cartan and a generator of a 
positive (resp. negative) subalgebra should be in positive (resp. negative) 
subalgebra, and brackets between pairs of generators, one from positive and one 
from negative subalgebra, should be in the Cartan.

Elements of the Cartan subalgra should be assigned zero weights. For a pair of 
generators from opposite subalgebras, if their weights do not add up to zero, the 
Lie bracket is automatically assumed to be zero. All other relations between pairs 
of generators from different subalgebras should be specified. The list of such 
relations should be either passed as the third argument to the function, if not, the 
function opens a dialog window to input relations between the generators. 

Below we build K2 Lie algebra, introduced by V.Kac

]. This Lie algebra has rank 2 free Lie algebras as positive and negative subalgebras,
glued together with a 2-dimensional Cartan subalgebra. It is a Kac-Moody like 
algebra, but without Serre's relations, and it corresponds to Cartan-like matrix [[2, 
2], [2,2]].

generators(Kpos,[E[1],E[2]],[[1,0],[0,1]]);

generators(Kneg,[F[1],F[2]],[[-1,0],[0,-1]]);

generators(Cartan, [H[1],H[2]],[[0,0],[0,0]]); store([H[1]&*H
[2]=0]);

# triangular(K2, [Cartan, Kneg, Kpos]); # This will use a 
dialog to enter the relations

triangular(K2, [Cartan, Kneg, Kpos], [H[1]&*E[1]=2*E[1],H[1]&*E
[2]=2*E[2],H[2]&*E[1]=2*E[1],H[2]&*E[2]=2*E[2],H[1]&*F[1]=-2*F
[1], H[1]&*F[2]=-2*F[2], H[2]&*F[1]=-2*F[1], H[2]&*F[2]=-2*F
[2], E[1]&*F[1]=H[1], E[2]&*F[2]=H[2]]);

lsimplify(E[1]&*F[2]);
0

lsimplify(((E[1]&*E[2])&*E[1])&*(F[1]&*F[2]));

Affine Lie algebras

Even though affine Lie algebras belong to the class of Kac_moody algebras, setting 
them up with Serre's relations is computationally inefficient. It is much more 
effective to use their realizations as central extensions of loop Lie algebras. Below 
we show how to use affine function to set up an untwisted affine Lie algebra of 
type G(2) . This construction is very similar to the construction of finite-
dimensional simple Lie algebras, except that each basis element has an extra (last) 
index that indicates the power of the loop parameter, and  there is a central 
element K.

affine(affG2, G[2]);

lsimplify(e[1,1,2]&*e[-2,-1,1]);

lsimplify(e[3,2,7]&*e[-3,-2,-7]);

It is possible to customize the names of the generators of affine algebra, and more
importantly their weights, by using additional arguments for this function. Here is
an example of the same affine G[2] algebra with different generator names:
affine(AFFG2, G[2], [E,H,K]);

lsimplify(E[1,1,2]&*E[2,1,3]);

Here is the same affine G[2] algebra, but with a Z-grading, which is more natural 
in the context of vertex algebras:
affine(affG2Z, G[2], [[k],[k],[0]]);

wt(e[3,2,7]);

Another important grading on affine algebras is the Kac-Moody grading, for which
all components are either non-negative, or all non-positive:
affine(affG2KM, G[2], [[r1+3*k, r2+2*k, k], [3*k, 2*k, k], [0,
0,0]]);

Here is the alpha_0 root vector:
wt(e[-3, -2, 1]);

Universal enveloping algebras

Given a Lie algebra L, we can construct its universal enveloping algebra U_L. All 
computations in the universal enveloping algebra are carried out in the Poincar -
Birkhoff-Witt basis, associated with the total order &< on the basis elements of the
Lie algebra. We use operation &@ to denote associative product. Associative powers
of elements are denoted by &^; for example, x&^3 is a short form for x&@x&@x.To 
bring an element of the universal enveoping algebra to its canonical PBW form, use 
function asimplify.

unienv(sl2);

asimplify(e&@h&@f&@e);

simple(sl3, A[2]);

unienv(sl3);

Casimir element
Casimir:=e[1,0]&@e[-1,0]+e[0,1]&@e[0,-1]-e[1,1]&@e[-1,-1]+e[-1,
0]&@e[1,0]+e[0,-1]&@e[0,1]-e[-1,-1]&@e[1,1]+(2/3)*h[1]&@h[1]+
(1/3)*h[1]&@h[2]+(1/3)*h[2]&@h[1]+(2/3)*h[2]&@h[2];

Casimir element is central
asimplify(Casimir&@e[1,0]-e[1,0]&@Casimir, U_sl3);

0

Verma modules

Given a Z-graded algebra, we can construct a highest weight Verma module using 
function hwrep. In the example below the arguments to this function are: M - name
of the module, sl2 - name of the Lie algebra, v - name of the highest weight vector,
[h=5] - list of rules indicating how elements of weight zero act on the highest 
weight vector. In this instance, we constructed a Verma module with the highest 
weight 5. The action of the universal enveloping algebra on the module is still 
denoted by &@. To bring an element of the module to its canonical form, use 
function repsimplify.

hwrep(M, sl2, v, [h=5]);

repsimplify(f&@v[5]);

repsimplify(h&@v[5]);

repsimplify(e&@v[5]);
0

repsimplify((e&^2)&@(f&^4)&@v[5]);

repsimplify(e&@(f&^6)&@v[5]);
0

Verma modules for algebras with a tringular decomposition 

Verma modules may be defined for any Lie algebra with a triangular 
decomposition, that is satisfying the condition that weights of element either have 
all non-negative or all non-positive components. This condition ensures that 
weight spaces in the Verma module are finite-dimensional (of course we assume 
here that graded components of the Lie algebra are finite-dimensional). Note that 
the construction of the highest weight module depends not only on Lie algebra, but
also on the grading imposed on it. The grading should be chosen in a way that 
there will be no Lie algebra elements having weights with mixed signs. Below we 
construct the Verma module for affine Lie algebra of type A[1] with the Kac-Moody
grading.

affine(affsl2KM, A[1], [[r1+k,k], [k,k], [0,0]]);

hwrep(MaffKM, affsl2KM, v, [h[1,0]=0, K=1]);

using();

repsimplify(e[-1,0]&@v[0,1]);

repsimplify(e[1,-1]&@v[0,1]);

repsimplify(e[1,0]&@v[0,1]);
0

repsimplify(e[1,0]&@e[-1,0]&@v[0,1]);
0

repsimplify(e[-1,1]&@e[1,-1]&@v[0,1]);
v

In this example we construct a Verma module for affine A[1] Lie algebra with the 
Z-grading, where we define the highest weigth with a function, rather than with 
the list of rules.
affine(affsl2, A[1], [[r1+2*k], [2*k], [0]]);

wt(h[1,3]);

wt(e[-1,1]);

unienv(affsl2);

mu := proc(x)
  if (x = h[1,0]) then RETURN(0); fi;
  if (x = h[2,0]) then RETURN(0); fi;
  if (x = K) then RETURN(1); fi;
end proc;

hwrep(Maff, affsl2, v, mu);

using();

repsimplify(e[-1,0]&@v);

repsimplify(h[1,0]&@v);
0

repsimplify(h[1,1]&@v);
0

repsimplify(h[1,-1]&@v);

repsimplify(e[1,-1]&@v);

wt(e[1,-1]);

repsimplify(e[1,1]&@e[-1,-1]&@v);
v

Verma modules with generic highest weights

We can also use indeterminants in the highest weight. These indeterminants, 
however, should be declared as consants in order for this to work.

constants:=constants, s, r;

unienv(Vir);

hwrep(V, Vir, u, [z=s, e[0]=r]);

repsimplify(e[2]&@u[s,r]);
0

repsimplify(e[-1]&@u[s,r]);

w := repsimplify((e[-1]&^2)&@e[-2]&@u[s,r]);

repsimplify((e[2]&^2)&@w);

repsimplify(e[1]&@u[s,r]);
0

repsimplify(e[2]&@e[-2]&@u[s,r]);

Singular vectors

We can search for singular vectors in a highest weight module. These are 
homogeneous (with respect to a given Z-grading) vectors that are annihilated by 
the positive subalgebra. Singular vectors may be used to construct submodules and
quotients. For example, for the Virasoro algebra, every submodule in a Verma 
module is generated by singular vectors [B.L.Feigin, D. B. Fuks, Verma modules over
the Virasoro algebra,  Functional Analysis and its Applications, 17.3 (1983), 241
-242.]. Below we show how to find a singular vector in one of the minimal model 
modules for the Virasoro algebra. The arguments to singularvector function are the 
weight of the homogeneous component of the Verma module where we search for 
the singular vector, the list of homogeneous raising operators and (optionally) the 
name of the module. The function computes vectors in the specified component of 
the Verma module that are annihilated simultaneously by all raising operators 
from the list. These raising operators can be arbitrary homogeneous elements of 
the univesal enveloping algebra with non-negative weights, but typically are 
chosen to be the generators of the positive subalgebra in the given Lie algebra. For 
the Virasoro algebra, its positive subalgebra is generated by e[1] and e[2].

using(Vir);

hwrep(MM, Vir, v, [e[0]=-3/8, z=-2]);

singularvector([-4],[e[1],e[2]], MM);

u:=op(1,%);

repsimplify(e[1]&@u);
repsimplify(e[2]&@u);
repsimplify(e[3]&@u);
repsimplify(e[4]&@u);

0

Submodules and Quotients

We can use singular vectors to generate a submodule in a highest weight module 
and then take the quotient by this submodule. Let us do this for the basic module 
for affine A[1] algebra

using(affsl2KM);

hwrep(MaffKM, affsl2KM, v, [h[1,0]=0, K=1]);

submodule(SM, [e[-1,0]&@v[0,1], (e[1,-1]&^2)&@v[0,1]], MaffKM);

quotrep(Basic, SM);

repsimplify(e[1,-1]&@e[1,-1]&@v[0,1], Basic);
0

basis([-2,-2], MaffKM); nops(%);

6
basis([-2,-2], Basic); nops(%);

2
basis([-2,-2], SM); nops(%);

4

Implementation of vertex algebras

We can use highest weight representations to implement vertex algebras, 
computing their fields and n-th products. Since vertex algebras are infinite-
dimensional, we set up a cut-off N, and work within the first N+1 graded 
components of the vertex algebra. Then in each field we keep only the components 

that change the degree by at most N. For this reason, we will be indexing 
components of the fields by degrees, rather than by n-th products/powers of z.

Below we consider the universal enveloping vertex algebra for sl(2), set up its 
generating fields, as well as the Sugawara field, and check that Sugawara field 
satisfies the OPE relations of the Virasoro vertex algebra. 

N:=4;

affine(vertsl2, A[1], [[k],[k],[0]]);

constants:=constants, C;

hwrep(Vaff, vertsl2, v, [e[-1,0]=0, e[1,0]=0, h[1,0]=0, K=C]);

u:=op(2,%);

Let us set up the generating fields:
Field_e := Array(-N..N): Field_h := Array(-N..N): Field_f := 
Array(-N..N):
for i from -N to N do
   Field_e[i] := e[1,i];
   Field_h[i] := h[1,i];
   Field_f[i] := e[-1,i];
od:

Next let us introduce a procedure that computes normally ordered product of two 
fields. Its arguments are the two fields and the degree (=conformal dimension) of 
the first factor.
NOP := proc(X, Y, degX)
global N;
local i,j,Z;

Z := Array(-N..N);
for i from -N to N do
    Z[i]:=0;
end;

for i from 1-degX to N do
    for j from -N to N do
         if (abs(i+j) <= N) then
             Z[i+j] := Z[i+j] + Y[j]&@X[i];
         end if;
    end;
end;

for i from -N to -degX do
    for j from -N to N do
         if (abs(i+j) <= N) then



(20.7)(20.7)

> > 

(20.8)(20.8)

> > 

> > 

(10.3)(10.3)
> > 

(18.5)(18.5)

(17.9)(17.9)

> > 

(9.2)(9.2)

(16.13)(16.13)

(20.6)(20.6)

> > 

> > 

(14.1)(14.1)

             Z[i+j] := Z[i+j] + X[i]&@Y[j];
         end if;
    end;
end;

RETURN(eval(Z));
end proc:

Here is a procedure for differentiation of fields
Der := proc(X, degX)

global N;
local i, Z;

Z:=array(-N..N); 
for i from -N to N do
   Z[i]:=(-i-degX)*X[i];
end;

RETURN(Z);
end proc:

Next let us define the Sugawara field. Note that for component-wise arithmetic 
operations on arrays we should use operations with ~, for example +~
L:=(1/(2*(C+2)))*~(NOP(Field_e, Field_f, 1) +~ NOP(Field_f, 
Field_e, 1) +~ (1/2)*~NOP(Field_h, Field_h, 1)):

L[0];
1

omega := repsimplify(L[-2]&@u);

Domega := repsimplify(L[-3]&@u);



(20.8)(20.8)

> > 

(20.10)(20.10)

> > 

> > 

(20.11)(20.11)

> > 

(21.2)(21.2)

(20.14)(20.14)

(10.3)(10.3)

(20.15)(20.15)

> > 

> > 

> > 

(18.5)(18.5)

(20.12)(20.12)

(17.9)(17.9)

> > 

(9.2)(9.2)

(16.13)(16.13)

> > 

> > 
(21.1)(21.1)

(20.9)(20.9)

(20.13)(20.13)

(14.1)(14.1)

> > 

Let us check the  Virasoro OPEs:  omega_(0) omega = Domega,  omega_(1) omega 
= 2*omega, omega_(2) omega = 0, omega_(3) omega = K/2*1.
repsimplify(L[-1]&@omega - Domega);

simplify(%);
0

repsimplify(L[0]&@omega - 2*omega);

simplify(%);
0

repsimplify(L[1]&@omega);
0

repsimplify(L[2]&@omega);

simplify(%);
Central charge of the Virasoro is 3C/(C+2).

Characteristic p

All of the computations described in this package may be carried out in positive 
characteristic.

characteristic(5);

simple(sl3, A[2]);



(20.8)(20.8)

> > 

(21.2)(21.2)

> > 

(21.3)(21.3)

(10.3)(10.3)

> > 

(21.5)(21.5)

> > 

(18.5)(18.5)

> > 

(21.6)(21.6)

(17.9)(17.9)

(9.2)(9.2)

(16.13)(16.13)

> > 

(14.1)(14.1)

(21.4)(21.4)

hwrep(Verma, sl3, v, [h[1]=2, h[2]=3]);

singularvector([-2,-2], [e[1,0],e[0,1]], Verma);
This is in agreement with modular representation theory (J.C.Jantzen, Modular 
representations of reductive Lie algebras, Journal of Pure and Applied Algebra, 152
(2000), 133-185).

characteristic(0);

singularvector([-2,-2], [e[1,0],e[0,1]], Verma);


